
Introduction: 
Data cleaning is an essential component of data analysis. Household level data is a 
particularly good candidate to undergo a serious and detailed data cleaning process. To a 
great extent, this is due to the nature of the data collection and entry process, which is 
often a protracted endeavor involving several people in different stages. As a result, 
errors and inconsistencies are prone to occur.  
 
Simple errors can involve a wrong label or a wrong variable name and can be easily 
noticed and fixed. More elaborate errors include duplicate household identifiers, 
combining string and numeric entries in the same variable, applying zeros to indicate 
missing values and conversely applying missing values to indicate zeros. These more 
elaborate errors are harder to fix and often require going back to the original 
questionnaires and in some cases to the enumerators. 
 
The implications of leaving these errors uncorrected are several and often seriously 
impact the credibility and reliability of subsequent analysis conducted on the data. For 
instance, the lack of unique identifiers may lead to wrong merges, which will in turn 
wrongly combine the information of different households or individuals. From a 
statistical perspective, replacing missing values for zeros will have considerable 
implications on the mean and the variance of the variable and will, therefore, bias any 
related estimates.  
 
A number of different software exists to handle and process household data and perform 
cleaning and analysis, each with different strengths. While we are cognizant of this fact, 
the focus of this manual is on data cleaning using Stata. In particular, the manual centers 
on the cleaning of household data and draws extensively from household datasets to 
provide examples, and to illustrate techniques and issues. 
 
In order to provide a comprehensive view of the multiple stages involved in a rigorous 
data cleaning process, this manual is divided into four chapters. Chapter 1 starts by listing 
the terms and definitions which the user is expected to be familiar with. Then, it provides 
the motivation for using Stata in general but also for performing data cleaning in 
particular. It concludes with a description of the basic syntax and commands in Stata. 
Chapter 2 provides an overview of the strategies used in the cleaning of string variables. 
Chapter 3 focuses on numeric variables and the various ways to label, describe, clean, 
analyze, detect and correct problems.  Chapter 4 combines all the pieces together and 
presents a strategy to data cleaning.  

Chapter 1: Stata as a tool for data 
cleaning 
 

Pre-requisites to understanding this manual 
 
This manual aims to assist the reader in cleaning data using Stata, and we assume some 
terms and definitions are familiar to the users. Some of these terms are listed below: 
 

a) Overall terminology: household, unique identifier, missing values, outliers, string 
variables, numeric variables.  

b) Descriptives: mean, median, variance, standard deviation, skewness, kurtosis, 
percentiles. 

c) Estimation results: coefficient, standard error, p-value, t-stats, z-stats, null 
hypothesis (rejecting and accepting), means testing, R2 

 
Throughout the text, definitions listed above will be placed as reminders, but it would be 
to the advantage of the user to become familiar with these terms beforehand. 
 
We also assume that the user will be writing ‘do-files.’ See below. 
 

Understanding and using do-files 
 
A do-file is a set of codes that tell Stata what commands to run. Other software call it 
programs, Stata calls them do-files. The name stems from the extension of the file, which 
is essentially a text file, saved as filename.do. It is an essential component of using Stata 
and of data cleaning because in addition to writing commands in it, the user can also 
document, make notes, state hypothesis and even make notes of some of the preliminary 
results. Another key advantage is that, as long as you do not add a command to explicitly 
save the data, nothing you do is irreversible. That is, if you made a mistake and deleted 
observations, as long as your file did not include a ‘save’ command, you can just re-run 
the file and get back to the point right before you made the mistake. At the data cleaning 
stage or during any analysis, a well-structured do-file also ensures that replicability can 
be achieved at a later time just by re-running the do-file. However, to ensure replicability, 
there are a few golden rules that should always be observed: 
 

1. The raw data should always remain untouched:  A lot of effort, time and money is 
often allocated to collecting household data. Thus, preserving the original dataset 
intact is essential not only to enable replicability but also to allow for cross-
checks once the data have been processed. In practical terms, this means two 
things: 

a. Keep a copy of your raw data in a separate place (drive, cloud, cd) where 
you are sure it will not be manipulated, saved over, or deleted.  A useful 
tip is to use the cd command to change the directory where processed files 

will be saved and call the raw data by typing the use command with the 
full path indicated.  
Example:   
cd c:\data\processed_files 
use c:\data\raw_data\file1 
*Perform manipulations 

 
save file1_clean  

 
This set of commands will tell Stata that any saved files should go to the 
directory \data\processed_files (cd=change directory), and that it should 
load the file from \data\raw_data (use). The save command, when used 
alone and not with a specific pathway, will save the file in Stata’s data 
format (.dta) in the directory specified through cd.  
 

b. Ensure that every thing done to your raw data files is outlined in the do-
file. It is often tempting to use the interactive feature of Stata (ie. the 
menus) to perform some simple manipulations. This, however, should be 
avoided particularly if it modifies the data. If you are simply running a 
tabulation command (tabulate) or trying out a regression model, then it is 
fine to do it interactively. However, if you use commands such as replace, 
generate, egen, collapse, contract or any other commands that modify the 
data, it is essential that you record these changes in the do-file.   
Example: 
use hhassets, clear 

  *Below we check for uniqueness of the household identifier 
  isid hhno 

/*There is an error on the identifier of the second household, which we 
correct below*/ 

  replace hhno=1324563 if hhno==13245633 
 
  If you perform this change without using the do-file, the next time you run  

the do-file the change will not be effected. 
 

c. Never save over the raw data. If, in the process of cleaning, you need to 
make changes, replace incorrect values or string with correct ones, ensure 
that you save the files with a new name, preferably in a different location. 
This can be easily ensured by following point (a). 
 

2. Document your do-file in a way that someone else would be able to understand 
what was done months or years later.  

a. It is always best to think about writing enough notes such that a random 
person who is unfamiliar with the project could open your file and 
understand what you did to the data and why.  

b. When working on a dataset, we take a number of steps to both clean and 
analyze the data. In some cases, these steps can be traced just by looking 

at the command. For instance, if we use the command table in order to 
generate a table, we know immediately that the purpose in applying the 
command was to generate the table. Suppose, however, that our analysis 
requires that we convert all our quantity variables (e.g. crop production) to 
kilos.  While we may remember why the conversion was done two weeks 
from the date we performed it, we may not remember it two years later. A 
random person reading your do-file also would not understand why you 
performed the conversion, and unless you specified *converting to kilos in 
a comment, it is unlikely they would be able to decipher what the math 
was. Thus, including a note in the do-file as to why the conversion was 
performed will save time and headaches in the future. The same applies to 
any other changes in the dataset.  

3. Your do-file or set of do-files should always enable you to start from the raw data 
and end with the finished product, whether it is a cleaned dataset or a set of 
regression results. It is helpful to see do-files as maps that guide us from the raw 
data to the final output. With that in mind, ensure that you add all the proper signs 
on the way to your final product.  Comments and notes can be added easily to do-
files without disturbing the sequence of commands. To insert a comment, just add 
an asterisk before the text (as was shown in the examples of 1.a. and 1.b. above). 

 
A well organized do-file ensures that current and future users of the dataset will readily 
understand the procedures and manipulations applied. In the process of generating a final 
product, however, it is often the case that intermediate files are generated for a variety of 
reasons. For instance, it may be necessary to generate a file containing total usage of 
fertilizer by region, which will be used to create a graph or make a table.  Files that are 
generated with a specific temporary purpose do not necessarily need to be kept alongside 
other files of the household survey.  Two possibilities arise for avoiding and/or dealing 
with the problem of unnecessary files. The first one is to use a temporary file. This option 
is used by advanced Stata users and is not covered in this manual, but the reader is 
encouraged to research about it using the help command or in the pdf documentation 
(help tempfile).  The second option is to state to erase the file using the erase command. 
Independent of which option is taken, both should be stated in the do-file. 
Example: erase c:\data\hhdata\production.dta 
 
Advanced Stata users also find it helpful to state the version of Stata being used at the 
time of the do-file’s creation. This is done with the version command and has the purpose 
of ensuring that the do-file will run smoothly in future versions of Stata, should 
commands change syntax or be eliminated or replaced. Often, the version command is 
placed on the top of the do-file prior to any coding.  
Example:  version 13  
 

Why Stata? 

 
Stata is currently one of the most used statistical packages available. A large number of 
universities, multilateral organizations (such as the World Bank) and research agencies 

(such as IFPRI) in the US and elsewhere have migrated to Stata. In fact, most US schools 
now prefer Stata over any other package. This means that using Stata will at the very 
least facilitate exchanges and collaboration with one or more of these institutions.  
Primarily, however, there are two main reasons why Stata is an appropriate tool. 
 
First, Stata is incredibly powerful. Stata has an enormous array of econometric routines 
already included in it, which makes various types of statistical analysis possible. In 
addition, Stata users (along with the Stata corporation) update Stata frequently by 
providing new codes, new estimation commands as well as fixing old ones. This, of 
course, suggests that anyone can create your commands in Stata and thus have the 
flexibility to clean and analyze data in a way that is tailored to specific needs.  
 
Second, Stata is very easy to use. Despite its richness in commands, it has a very simple 
and straight-forward language. There are three different ways of using Stata:  
 

• By pointing and clicking using the pull-down menus 
• By entering commands on the command window and using it interactively 
• By writing commands to a do-file (recommended)  

 
The syntax for writing commands in do-files or interactively is straight-forward (see 
below). 
 

Why use Stata for data cleaning? 
 
These general features of Stata are of interest to analysts performing analysis and/or data 
cleaning.  There are many features that are particularly suitable to data cleaning, which 
can expedite the process of cleaning as well as to provide the analyst with the tools 
necessary to undertake the cleaning in an accurate way.  Some of them are listed below: 
 

The ability to quickly visualize your data 
You can visualize the distribution of your data very quickly with just a one-line 
command.   For instance, if we wanted to see the spread of the age variable across all 
households we could simply type:  
 
histogram age  if  age>0, norm 

Figure 1 - Distribution 

 
The plot shows a distribution that is highly skewed, indicating the prevalence of young 
individuals. If we are not familiar with the age structure of the individuals in our dataset, 
we may be intrigued by the large number of young people. Could it be that the ages were 
wrongly entered or is it just that the households have many kids? A useful check may be 
to look what happens when we plot the distribution of heads of household only. To do 
that, we just need to complement the previous “if” statement. The variable relhh listed 
below relates to the relative position of the member in the household (here it is useful to 
already note the naming convention – rel indicates relative, hh is household, so the 
variable name for relative household position is named relhh). Member number 1 is the 
head of the household. 
 
histogram age if  age>0 and relhh==1, freq norm 
 

Figure 2 – Conditional distribution 

Now we can be reassured that in fact our dataset is fine. Heads of households are all 
within the expected range of age, i.e. there are no kids as head of households.  
 
The ability to perform calculations across many variables 
 
You can extract important descriptive statistics, create new variables, and perform 
calculations across several variables with one command. This process is called looping 
and can be performed in a number of different circumstances. For instance, suppose we 
have four variables on quantities sold of different inputs. We know that we cannot have 
negative quantities sold. We can verify if negative values exist quickly by using a loop 
and the assert command: 
 
foreach var of varlist maizeq wheatq soybeanq riceq { 
assert `var'>=0 
} 
 
The command assert checks whether the logical statement holds. If the statement does 
not hold, an error is displayed1. The “loop” part of the command is introduced with the 
foreach command. It is possible to check numerous variables very quickly (see section 
called “Looping in Stata”). Briefly, the syntax is the foreach command, then var refers to 
the temporary local macro you will be using inside the loop, of varlist indicates you are 
about to list the variables that should be looped through, and maizeq wheatq etc are the 
variables. What Stata will do is perform whatever is within the loop brackets, {  } as 
many times as there are variables. In this case, it is the assert command with each of the 
variables. That is, this loop is the equivalent of writing: 
 
assert maizeq>=0 
assert wheatq>=0 
assert soybean>=0 
assert riceq>=0 
 
Clearly, it is much faster to learn to use loops as soon as possible as they are some of the 
most flexible and useful tools in Stata. Let’s see another example: the same strategy can 
be used to, say, generate new variables. Suppose we wanted to standardize (transform all 
variables to have a mean of zero and standard deviation of 1) our variables because we 
suspect the variables are expressed in different units. 
  
 foreach var of varlist maizeq wheatq soybeanq riceq { 
egen st(`var’) >=std(`var’) 
order st`var’, after(`var’) 

                                                 
1 The fact that Stata displays an error when the assertion is wrong is very helpful during the process of data 
cleaning as it forces us to check any problems that might be causing the error. However,  in some cases 
errors in the assertion may be expected or may be correct in the command that follows the assert command. 
In such cases, using the option rc0 after the assert command is helpful as it displays the error but does stop 
the do-file from running. 

} 
 
One very important note when using local variables such as var, you need to use the 
appropriate single quotes or your command will not work. Note the quotes used are not 
two right quotes, such as 'var', but one left and one right quote, like so: `var'. In an 
English keyboard, the left quote is in the top-left under the tilde, and the right quote is 
next to the return/enter key. Whatever your keyboard, it is likely these quotes are two 
different keys you will have to use. Make sure you familiarize yourself well with them. 
 

The ability to write your own commands 
 
Once cleaning is finished and you want to analyze the data, you will find a formidable 
suite of statistical capabilities, ranging from standard Ordinary Least Squares regressions 
to count models, to dynamic models using generalized methods of moments, to 
sophisticated panel data models. In addition to using the built-in commands, you can also 
program your own commands. Below is an example of a program that runs an OLS 
regression using Maximum Likelihood. 
 
program mynormal_lf2 
version 12.0 
args lnf mu lnsigma 
quietly replace `lnf' = ln(normalden( $ML_y1, `mu', exp(`lnsigma'))) 
end 
ml model lf mynormal_lf2 (yield = fertilizer labor capital) () 
ml maximize 
 

The ability to get help when you need it 
 
All manuals of Stata are built into the software since Stata 11. Thus, if you would like to 
use the regress command but do not know how or have forgotten, you can simply go to 
the help menu and click on pdf documentation.  There you will find not only the syntax 
for the command but also the statistical assumptions that underpin it, the different ways 
the variance can be estimated and the relevant academic references.  In addition to the 
help command, there are two other commands that provide assistance: search and findit. 
 
The search command provides a broad way of looking for material related to a particular 
topic or even a particular author. It pulls together all available material on the topic and 
presents it to the user on the screen.  If we type search panel data, Stata will present a 
screen with the panel commands, as well as Stata Journal issues and other web links 
related to the topic. It also displays links to download user-written commands. We could 
also search for any references written by a particular author, e.g. William Gould (Stata’s 
president and developer). To do so, we just need to add the option  author.  search  
William Gould, author  

The findit command has become superflous starting in Stata 13. The search command 
now performs the actions findit used to perform, i.e. find user-written commands and 
references on a particular topic.  Its syntax is similar to the search command. Typing 
findit panel data will provide the user with a screen with a number of references and 
user-written commands for panel data. 
 

Overview of Stata’s syntax and basic commands 
 
The vast majority of commands2 in Stata follow a fixed syntax. The order of the 
command is usually:  
 
command variables [if/in conditions] [weights], options  
 
Stata will run with just the command and the variable list. The if/in conditions, the 
weights and the options can be used as required by the analysis. Let’s start with an 
example using just the command and the variable list. 
 
regress yield labor capital 
 
The regress command is just followed by the variables. Suppose that we would like the 
regression to include only households that are located in the Northern part of a given 
country. To do so, we can type: 
 
regress yield labor capital if region==”North”3 
 
or the first 100 households in the dataset 
 
regress yield labor capital in 1/100 
 
As many “if” conditions can be added as needed.  “if” conditions can be separated by two 
logical connectors: and, or. The first one --“and”-- is expressed by an ampersand, while 
the second one is expressed by a vertical bar, as illustrated in the example below 
 
regress yield labor capital if (region==”North” | region==”South”) & year>1980 
 
                                                 
2 Stata allows commands to be abbreviated up to a point where it can almost unambiguously understand the 
command. For instance, we can type reg instead of regress, u instead of use, gen (or even g) instead of 
generate. However, we cannot type re for regress because other major commands have the same prefix 
(such as replace and restore). Stata shows the possible abbreviation of each command on the help screen 
by underlining the part of the command that can be used as an abbreviation. Variable names can also be 
abbreviated as long as the abbreviation does not lead to ambiguity with other variable names. For instance, 
if a variable is called income and there are no other variables with the same prefix, we can abbreviate it by 
calling it inc. However, if we had two variables, one called income_male and the other called 
income_female, the previous abbreviation would not work as Stata would not know which variable to use.  
3 Note that one equal sign, =, is used for definitions. That is, if you type generate region=”North”, it will 
assign the value of “North” to the variable region. If you are comparing, such as using an if statement, then 
it is a double equal sign you need, such as if region==”North”. 

This statement is telling Stata to perform an OLS regression (regress) the variable yield 
(dependent variable) on variables labor and capital (the independent variables) for the 
regions ‘North’ and ‘South’ (both regions included as we used “or” condition) and for 
years greater than 1980 (“&” means both conditions have to be satisfied). 
 
The or and and conditions are logical statements that tell Stata how to subset the data for 
the desired analysis. Or suggests that Stata can use observations from all groups present 
in the condition. In the example above, we are telling Stata that all observations from 
regions North and South are to be analyzed. And suggests that two or more conditions 
need to be met simultaneously. The example above illustrates that two conditions have to 
be met: First, the regression is to be performed for regions South and North and second, 
only data points for which the year is greater than 1980 should be included.  
 
Note that parenthesis were added to the “or” condition. The parenthesis are needed every 
time “and” and “or” are combined in a multiple condition statement. Stata will not 
prompt an error if parentheses are not included, but may deliver results that were not 
intended.  
 
Weights can also be added to commands.  Stata has four different types of weights: 
frequency, analytic, sampling and importance.  An excellent description of these weights 
is provided on the following website: 
http://www.cpc.unc.edu/research/tools/data_analysis/statatutorial/sample_surveys/weight
_syntax 
 
For the purposes of household data, the most important weight is the sampling weight (or, 
probability weight). The sampling weight is defined is the inverse of the sampling 
fraction. Thus, if a household has chance of being selected of 1 in 5 (1/5), the sampling 
weight is 5. It is essential to apply the sampling weights when estimating parameters 
from survey data. Sampling weights affect both the mean and the standard errors of 
parameters.  To apply sampling weights (or any other weights) in a command, all that is 
needed is to indicate the type of weight and then the variable that contains the weight. For 
instance, below we run a regression with the sampling weights. Weights are specified 
before any options (i.e. before the comma).  
 
regress yield labor capital if region==”North” [pw=weight] 
 
Stata has a complete suite for estimating models and parameters from survey data. Survey 
commands are preceded by the prefix st.  The reader is encouraged to study the Survey 
data manual provided on the PDF documentation that comes with Stata. 
 
If we suspect that the error term of our regression may be heteroskedastic, we may want 
to use robust standard errors in our estimation. To do so, we simply add the option robust 
 
regress yield labor capital if region==”North”, robust 
 

http://www.cpc.unc.edu/research/tools/data_analysis/statatutorial/sample_surveys/weight_syntax
http://www.cpc.unc.edu/research/tools/data_analysis/statatutorial/sample_surveys/weight_syntax

Options are always preceded by a comma. Once a comma is placed, options can be listed 
in sequence and do not need to be separated by additional commas, as demonstrated 
below.  
regress yield labor capital if region==”North”, robust noconstant 
 
Most commands also allow the use of “by” before the command line to tell Stata that 
estimations or tabulations are to be done by categories. In such cases, the command 
above becomes: 
 
by region: regress yield labor capital, robust noconstant 
 

The basic commands 

Commands to access your data 
 
If your data is already in Stata format you can access it in two ways:  

1) by using the pull-down menus. The tab to open a new file is called file. Within 
file tab, the option Open is displayed, as shown in the screenshot below.  

2) If you would like to open the file from your do-file (preferred), you can also type: 
 

use mydata, clear  
 
The option clear after the comma tells Stata to clear the memory so that other 
files can be loaded. This option is not mandatory but facilitates the flow of the 
program as it keeps Stata from displaying the error message “no, data in memory 
would be lost”. The user may also simply type clear prior to accessing a new 
file. 

Figure 3 – Stata environment (Windows) 

Note that if you choose the second option you need to make sure you are in the right 
directory. If you are not, you can change the directory using cd or inform it in the use 
command. There are two ways of informing Stata where your file is: 
 
The first is by providing the directory information in the use command, see example 
below. Note, however, that doing via the use command does not change your directory 
 
Example: use c:\data\myfiles\hhincome, clear 
 
Alternatively, you can also change the directory by using the cd command as illustrated 
below. Once you change the directory, you do not need to provide the directory in the use 
command and can simply type the file name. 
 
cd c:\data\myfiles 
use hhincome, clear 
 
If your data are not already in Stata format (extension .dta) there are a few options: 
 

• Use software such as Stat Transfer to convert it to Stata. 
• Use the insheet, infile, or infix commands. These commands allow you to specify 

which variables to be imported into Stata from text or spreadsheet file types (such 
as comma separated values), including their format and their length. The User 
Manual (page 309) contains details about these commands. 

• If data are in a spreadsheet format (such as data created using Microsoft Excel), 
import it using Stata’s import command. The import command was implemented 
in Stata with version 12 and is very easy and practical to use. The general syntax 
is: import using file.xlsx. Variables and options can also be specified. It is often 
also possible to copy and paste your data from spreadsheet software into Stata’s 
data editor, though this is not advisable, as you cannot put this in a do-file. 

 
Regardless of the method you select, there are a few reminders that should always be kept 
in mind: 
 

• Numbers and letters do not mix in the same variable. Numbers in a variable with 
even one letter will all be treated as letters (strings).  

• Use short variable names, preferably in lower caps. 
• No spaces in variable names. 

Commands to generate and change your data 
 
This section will provide a broad idea of the main commands to generate, change and 
save your data.  

generate 
 
The generate command allows you to create new variables and follows a slightly 
modified syntax. Generally, the syntax can take two forms: 
generate [type] x3=x2+x1 [if/in]  
 
In this case the command is being used to generate a variable that is a linear combination 
of other variables. Transformations to the variables can also be performed by adding 
some of the functions available in Stata, like so: 
 
generate x3=sqrt(x2)+ln(x1) if x1<=100 
 
generate u=runiform()  
 
creates an uniformly distributed random variable. 
 
The command generate has a companion command called replace. Because variables can 
only be created once, any modifications to the created variable(s) need to be made using 
the command replace.  In the example above we created variable x3, which is non-
missing for all values of x1<=100. If we would like x3 to take different values when 
x1>100, we would use replace: 
 
replace x3=x2^2 if x1>100 & x1!=.  
 
Note that we added an additional “if” condition”, specifying that x1 also needs to 
different than missing. The reason for doing so is because missing values are very large 
numbers in Stata. Hence, had we not excluded missing values from the calculation and 
there were missing values in x1, Stata would have created values of x3 when x1 was 
missing (which in most cases is not desirable).  
 
egen 
 
The command egen is an extension to the command generate. It allows you to create 
variables by performing functions. For instance, egen can be used to create several 
descriptive statistics across rows, including: mean, median, sum, min, max, skewness, 
kurtosis, inter-quartile range and many others. The syntax is similar to generate except 
that egen always requires a function to be specified. 
 
egen minvar=rowmin(yield1 yield2) 
 
This tells Stata to create variable minvar with the minimum value of the row with the data 
yield1 and yield2. You can also count the number of missing and non-missing values, 
concatenate string variables, generate ranks, create numerical identifiers for categorical 
variables, obtain non-parametric statistics, and fill in patterns among many other uses. 

collapse 
 
The collapse command provides a quick and useful (though dangerous) way to obtain 
summarized statistics. It is dangerous because it changes the data in a non-reversible way. 
collapse allows you to transform your data quickly. It can be used to aggregate data and 
change the structure of the dataset.  Suppose we have a dataset on age and years of 
schooling by gender for a number of different students. To obtain the average age and 
average schooling by gender, we can type: 
 
collapse (mean) age schooling, by(gender)   
 
More than one statistic can be used with collapse. In the example above, we can add the 
standard deviation: 
 
collapse (mean) age schooling (sd) agesd=age schoolingsd=schoolsd , by(gender)   
 
It is important to realize, however, the irreversibility of this command. Based on the 
options specified, the command changes your dataset—thus care is needed when using it 
and particularly if it is followed by the save command. This point cannot be overstated: 
whenever saving your data following a major change, it is always a good idea to use 
a different name. Alternatively, you may wish to use the very useful preserve and 
restore commands before and after the command collapse. This set of commands are 
discussed in chapter 3. 
 
contract  
 
The command contract performs the same function as collapse, but only with 
frequencies. 
 
reshape  
 
The command reshape also allows you to change your data but in a different way. Unlike 
collapse and contract, reshape is reversible. If you are performing a reshaping that is not 
reversible, Stata will let you know and confirm you want to do it. If this is the case, it is 
advisable to follow the rules for collapse as well.  
 
Reshape is used to transpose the data. Using it is appropriate when the analysis desired 
requires the data to be in a different format.  For instance, for panel data analysis we 
often need to have variables stacked (referred to as “long” format) and not spread out 
across columns (referred to as “wide” format). The help function in Stata is very useful 
for this command, particularly as it gives you a quick example of data in long and wide 
formats to make sure you are assigning the proper variables to the proper spots. In 
addition to help function in Stata, the following websites detail the use of the reshape 
command. (These are also discussed further in chapter 3.) 
http://www.ats.ucla.edu/STAT/stata/modules/reshapel.htm 

http://www.ats.ucla.edu/STAT/stata/modules/reshapew.htm 
 
save 
 
The command save saves the file and is self-explanatory: 
 
save filename 
save filename, replace 
 
It is always useful to add ,replace after save if you are including this in your do-file, as 
the do-file will stop running if the file already exists. However, if the file does not exist 
Stata will simply note the file did not exist but will not break the program’s run. Be sure 
that you are including different file names when you make significant changes to the 
dataset, however!  
 
drop and keep 
 
The commands drop and keep are also self-explanatory. They drop or keep variables. 
These two commands perform the same action but from different angles. Suppose we 
have a dataset with the following variables: x1, x2, x3, x4, x5, x6, x7, x8. For the 
purposes of the analysis, we only need x2 and x3. There are two ways of doing it, using 
the drop and keep commands:  
 
drop x1 x4 x5 x6 x7 x8  
 
Or 
 
keep x2 x3 
 
In this example using keep makes more sense. However, if instead of keeping x2 and x3 
we wanted to keep x4 x5 x6 x7 x8, drop would be more sensible. Which one you use 
depends on which list is longer – what you want to keep or what you want to drop. 
 

Commands to describe your data 
 
The commands in this section provide a broad view of the data and are often used in the 
first stages of cleaning or analysis. 
 
describe 
 
The command describe gives you an idea of the variables (their types) and their 
descriptions (if available). Using describe will describe all variables in the dataset, but 
you can also choose which variables you would like. 
 
describe age income 

This will tell Stata to describe only these two variables, rather than all the variables in the 
dataset. 
 
codebook and inspect  
 
Both these commands provide you with general information about your variables and are 
good first steps to understanding your variables. 
 
codebook  age income 
inspect  age income 
 
list 
 
The command list gives you a list of the entries in your variables, but can also calculate 
some statistics and display them. 
 
list age income in 1/5 
list age income in 1/5, sum mean 
list age income in -10/-1, sum mean sepby(gender) noobs 
 
label 
 
The command label allows you to label your variables. Often variable names have to be 
short and succinct and this is where labeling becomes appropriate. You can specify what 
the variable is, the units, how it was calculated, etc. Proper labeling means an easier life 
in the future. The syntax is: label var variable_name “variable label” 
 
Here is an example of how to label a variable, using the variable income: 
 
label var  income “Per capita income in US$” 
 
Stata also offers value labels, which can be used to attribute labels to numeric data when 
the data are categorical. Variables can be of a binary nature or have multiple categories. 
Any variable that takes “yes” or “no” value is a good example. Typically, household data 
contains several variables that have values 0 for “no” and 1 for “yes”. Here are is an 
example of how to use the label command: 
  
label define yesno_label 0 “No” 1 “Yes” 
label values children yesno_label 
label values woman yesno_label 
 
Note that the once the label value is defined, it can be applied to more than variable. For 
instance, if in a dataset there are different variables that take on values of yes/no, the 
same label can be applied to all of them. In our example, we created a label that labels 
variables with value 0 as “no” and with value 1 “yes”. Then we assign this label to both a 
variable named children (which could be, for example, 1 if there are children in the 

household and 0 if not), and also assigned the value to the variable named woman (which 
we could say is a variable that takes on a value of 1 if the individual is a woman, or 0 if 
not). Both are labeled using the same label. 
 
isid 
 
The command isid tests whether a variable is uniquely identified. If the variables are not 
unique an error message is displayed. 
 
isid hhid 
 
duplicates 
 
The command duplicates checks if there are duplicates and Stata can report/tag/remove 
them easily. Duplicates can be used to identify duplicates in all variables of the dataset or 
in specific variables. In household4 data, typically the household identifier is expected to 
be unique. We covered why this is important earlier in the manual, for example, to 
facilitate merges across years and ensuring we have the right number of households and 
individuals we expect from our original sampling activities. 
 
duplicates report hhid 
duplicates tag hhid, gen(tagid) 
duplicates drop hhid, force 
 
The first command will report the duplicates in terms of the variable hhid. The second 
will identify the duplicates in hhid, creating a new variable, tagid and assigning to it a 
value of 0 if it is the first instance of that value in hhid, a 1 if it is the second instance, a 2 
if it is the third instance and so on. The last command forces Stata to drop the duplicate 
entries – use this command with caution. You cannot choose which ones are dropped if 
you use this command. 

Commands to analyze your data descriptively 
 
Commands in this section are the next step in the cleaning or analysis of a dataset. They 
provide summary statistics, tabulations and general descriptive statistics. 
 
table  
 
The command table is used to generate tables. One-way or two-way tables can be 
generated and incorporate a range of descriptive statistics. 
                                                 
4 Files from the household data can come in a variety of formats depending on the survey. Agricultural 
modules, for instance, can have several units of observations across different files: household, household 
and plot, household and individual, household and crop among others. Uniqueness must occur at the unit of 
observation whether household, household and individual or any other unit of observation. Note that in 
cases in which the unit of observation is made of two or more variables, the individual variables that 
compose the unit are not unique. Thus, to check for duplicates in such cases, the two variables must be 
included. For instances, duplicates report hhid memberid. 

table state, c(mean yield sum income p50 price) row col 
The variables after the table command are the classification variables, also known as row 
and column variables. These variables serve only as categories in which other variables 
will be analyzed. After the comma, a “c” appears followed by parenthesis. The “c” stands 
for contents and provides the statistics desired. If  “c” is not provided Stata assumes that a 
frequency count is requested. 
 
bysort state: table age_group, c(mean income sd income) 
 
The bysort command here sorts the data by state and then crates the tables by state. 
 
Sub-totals by rows or columns can be obtained by adding the appropriate options 
table age_group gender, c(mean income sd income)  row col 
 
tabstat 
 
The tabstat command is another way to make tables and is usually faster than the table 
command. Its syntax is also fairly simple, but slightly different from table. Desired 
statistics are listed after the comma and are preceded by the word stats. tabstat only 
allows for one class variable, which can be specified under the option “by”. If “by” 
precedes the command, more variables can be specified, as the example below shows: 
 
tabstat weight price income yield, stats(mean sd p50 skewness kurtosis) by(state) 
 
by state county: tabstat weight price income yield, stats(mean sd p50 skewness kurtosis) 
 
The second statement first sorts the data by state, then within state by country and 
subsequently performs the rest of the statement. 
 
tabulate 
 
The tabulate command allows you to obtain frequency counts as well as perform selected 
tests (chi squared, fisher tests etc) 
 
tab  gender age_group 
tab  gender age_group, chi //chi-squared test 
 
summarize 
 
The summarize command is self-explanatory. It summarizes the main statistics of the 
variable you choose, such as the mean and standard deviation. By adding the option 
detail you can get more measures, such as the percentiles, skewness and kurtosis. 
 
summarize age 
summarize age, detail 

total, mean and proportion 
 
These commands provide the sum, the mean and the share of variables in the sample 
along with the standard error. 
 
total  age 
mean age 
mean age income education, over(gender) 
proportion age income education 
 

Looping in Stata 
 
Processing one variable is easy, but what about 100? For example, checking whether 
crops 1 and 2 have yields greater than 10 tons per hectare only takes one command, but 
what about checking for crops 3 to 250?  There are many reasons why we need to 
perform loops (repeated operations that are done sequentially, conditional (or not) on 
some specific constraints) and Stata has a powerful suite to perform loops. Below we 
discuss two commands that can assist in the execution of loops: foreach and forvalues. 
 
foreach  
 
The foreach command allows the user to perform a number of different types of loops by 
making simple adjustments to the syntax. The basic syntax of foreach consists of listing 
the objects for which the loop is to be performed. After listing, a { is used to indicate the 
beginning of the loop. The lines that follow are the commands which Stata will execute in 
the loop. At the end of the commands, the } is closed. Below, we provide an example. 
 
foreach var of varlist x1 x2 x3 { 

replace `var’=. if `var’==-99 
} 
 
The first line of the command lists the variables of interest and in that process creates a 
macro called var. It need not be var , the user can choose any name. The word varlist is 
added because the loop is for variables. The second line applies the command replace to 
the `var’5  when the variable equals -99.  Stata understands the code above as an 
instruction to start with x1, perform the commands then proceed to x2 and than x3. 
 
The foreach command also allows us to: 
 

                                                 
5 Once created macros have to be referred to in simple quotes `  ’.  Failing to include the quotes will result 
in error because Stata will read the word var  and proceed as if var was the varaible name. For details on 
macros, the reader is advised to read the Programming manual of Stata, available in the PDF 
documentation provided with Stata. 

1. Run multiple do-files 
Ex:  foreach name in  file1 file2 fil3 { 
do `name’ 
} 
 

2. Perform operations across multiple files 
Ex. foreach file in file1 file2 file3 { 
rename income income_old 
drop age 
save `file’, replace 
}  
 

3. Specify numbers or ranges of numbers to be checked against one or more 
variables  
Ex. foreach k in 1 2 3 4 5 6 7 { 
drop if age==`k’ 
}  

 
forvalues  
 
The forvalues command follows a similar logic but is applied to numbers. Suppose we 
have 100 variables x1-x100 and we would like to generate new variables that are the 
natural log of our original variables. We can do that using the forvalues command: 
 
forvalues  i=1/100 { 

gen lx`i’=ln(x`i’) 
}    
 
The i specified in the code above is a macro just like the var in the foreach command.  
forvalues can also perform loops for different intervals. For instance, if we wanted to 
generate the natural logs only for variables x10, x20, x30… x100, we could have typed 
 
forvalues i=1(10)100 { 

gen lx`i’=ln(x`i’) 
}    

Accessing estimation results 
 
The commands mean, total and proportion provide an opportunity to discuss how we can 
obtain and use results from the analysis.  Often intermediary steps are required to obtain 
certain parameters (mean, standard error and others) in order to continue with the 
analysis. In simple cases, such as the mean for one variable, it is not a big problem to just 
copy the result and paste it in the do-file. However, in many cases, several results need to 
be used and copying and pasting them from the screen is not only time-consuming but 
also prone to errors. Stata saves selected results from a number of commands. These 
results are saved as macros and can be used to retrieve calculations performed by the 
user. 

For instance, suppose we have a yield variable with 100 observations and we want to 
obtain its mean. 
 
summarize yield 
 

 
 
In addition to displaying it, summarize stores this information as scalars in a macro. To 
see it, we type 
 
return list 
 

 
All scalars are presented as r(statistic). To bring these scalars into a calculation, we can 
just type the statistic of interest, e.g. r(mean). We may be interested in demeaning the 
variable yield (i.e. removing the mean from each observation) in order to understand how 
the variable varies around the mean.  To do so, we can simply type 
 
summ yield 
gen yieldct=yield –r(mean) 
 
The advantages of using a macro are several: first, Stata keeps the entire number (with all 
decimals) and therefore is more accurate; second, errors are avoided as no data entry is 
needed; third, we can use the same strategy in a loop to perform the calculation over 
different variables.  
 
Estimation commands, such as mean, regress and others store information in matrices, 
but can also be retrieved via macros. 
 
Note that to obtain estimation commands, we have used ereturn.  The list after ereturn 
presents all the information that Stata stores according to the type of information: scalars, 
macros, matrices and function. Accessing a scalar is similar to the procedure describe for 
summarize. To access the matrices involves a bit more effort. The coefficient can be 
accessed by typing _b[yield], preceded by the relevant command (display, generate, etc). 
The variance can also be accessed but first we need to transform it into a scalar. 

       yield         100    52.38146    29.57984    .673274   99.61973
                                                                      
    Variable         Obs        Mean    Std. Dev.       Min        Max

                r(sum) =  5238.146450340748
                r(max) =  99.61972808837891
                r(min) =  .6732739806175232
                 r(sd) =  29.57983774985505
                r(Var) =  874.9668013077499
               r(mean) =  52.38146450340748
              r(sum_w) =  100
                  r(N) =  100
scalars:

. return list

matrix v=e(V)  
scalar alpha=v[1,1] 
 
To see the standard error, we can just type 
display sqrt(alpha)  
 
The procedure performed above consists of three stages. The first part creates a matrix e 
which is equal to the variance matrix obtained from the estimation command.  The 
second line pulls out the variance (for simplicity, we just chose the first observation in the 
diagonal of the matrix) and puts it into a scalar (a number). We have chosen to call this 
number alpha, but it could be called anything.  The third statement displays the square 
root of the alpha, which is the standard error. 
 
In chapter 3 we explore this topic further by applying this knowledge in a context of 
multiple variables.  
 

Commands to combine datasets 
 
A key component of household data analysis and cleaning is combining files from 
multiple modules. Stata has two main commands that allows us to combine files: merge 
and append. 

             e(sample)   
functions:

              e(error) :  1 x 1
                 e(_N) :  1 x 1
                  e(V) :  1 x 1
                  e(b) :  1 x 1
matrices:

             e(depvar) : "Mean"
         e(properties) : "b V"
       e(marginsnotok) : "_ALL"
            e(varlist) : "yield"
          e(estat_cmd) : "estat_vce_only"
              e(title) : "Mean estimation"
                e(vce) : "analytic"
                e(cmd) : "mean"
            e(cmdline) : "mean yield"
macros:

               e(rank) =  1
               e(k_eq) =  1
                  e(N) =  100
             e(N_over) =  1
               e(df_r) =  99
scalars:

. ereturn list

                                                              
       yield     52.38146   2.957984      46.51218    58.25075
                                                              
                     Mean   Std. Err.     [95% Conf. Interval]
                                                              

Mean estimation                     Number of obs    =     100

. mean yield

The merge command combines files based on the relevant identifiers. Files can be 
merged with both unique and non-unique identifiers.  
 
The user should know whether the identifiers are unique or not. The table below shows 
the different syntaxes of the merge command depending on the uniqueness of the 
identifier 
 

Table 1 
 

Identifier in file 
1(master) 

Identifier in 
file 2(using) 

Syntax 

Unique unique Merge 1:1 id using file2 
non-unique unique Merge m:1 id using file2 
Unique non-unique Merge 1:m id using file2 
non-unique non-unique Merge6  m:m id using 

file2 
 
 
One-to-one merges 
To combine two datasets, Stata uses common information present the both datasets. This 
information is often referred to as identifiers. Stata can merge datasets that have different 
levels of uniqueness in the identifiers, as indicated in the table above.  When identifiers 
are unique in both datasets, the syntax 1: 1 is used to suggest that each observation in the 
variable listed as an identifier will find one and only one match in the second dataset.  
This would happen, for instance, if we wanted to combined files that contain household 
level information e.g. one file containing total income and another containing location of 
the household. 
 
Many-to-1 or one-to-many merges 
Often, we may be interested in adding information from the household to a file that 
contains data on individuals living in the household. In this case, we would resort to the 
m:1 syntax since the unique identifier of the individual level data arises from the 
combination of the household identifier and individual identifier.  In practice, this means 
that the household identifier in the individual level file will be repeated, while the 
household identifier in the household level file will be unique.  The many to one (m:1) or 

                                                 
6 Many to many merges are dangerous and may lead to unpredicted results. The text below, taken for the 
Stata manual lists some of the dangers and suggests that it should not be used. “m:m   specifies a many-to-
many merge and is a bad idea. In an m:m merge, observations are matched within equal values of the key 
variable(s), with the first observation being matched to the first; the second, to the second; and so on. If the 
master and using have an unequal number of observations within the group, then the last observation of the 
shorter group is used repeatedly to match with subsequent observations of the longer group. Thus m:m 
merges are dependent on the current sort order—something which should never happen. 
Because m:m merges are such a bad idea, we are not going to show you an example. If you think 
that you need an m:m merge, then you probably need to work with your data so that you can use a 
1:m or m:1 merge.” Stata, 2013. 

one to many merges (1:m) cover these possibilities. Which one you use depends on 
which dataset you have loaded on your screen and which one you are calling to be 
merged into the one you have open. 
 
Many-to-many merges 
Although this is an option in Stata, it should be used with caution (if ever). It is generally 
best to make sure there is a unique identifier in at least one of your files, and in that way 
you can track how your data is being merged. The many to many merge makes this hard 
to follow properly. 
 
The output of the merge command provides the user with a clear idea of the success of 
the merge, as well as the source of every observation in the new combined file. An 
example is provided below: 
 

 
 
The example shows the number observations that were perfectly matched at the bottom. 
On the top, non-matched observations are presented broken down by source: from master 
means that observations exist only in the master dataset (i.e. the dataset you are running 
the merge command from); from using means that observations exist only in the using 
dataset. In addition to displaying this table, Stata also creates a variable called _merge, 
which allows the user to select which observations to keep.  
 

 
 
 
If we wanted to keep only the perfect matches, we would type: 
 
keep if _merge==3 
 
Merge has a number of other options which the user can learn about by typing help 
merge. 
 
 
 
 

                                             
    matched                            12,480  (_merge==3)

        from using                        896  (_merge==2)
        from master                       897  (_merge==1)
    not matched                         1,793
                                             
    Result                           # of obs.

                  Total       14,273      100.00
                                                            
            matched (3)       12,480       87.44      100.00
         using only (2)          896        6.28       12.56
        master only (1)          897        6.28        6.28
                                                            
                 _merge        Freq.     Percent        Cum.

. tab _merge

append 
 
The append command combines files by stacking them. If file 1 has price data for 20 
crops and file 2 has price data for another 10 crops, we can combine the two files using 
the append command. With file 1 loaded in memory, we type 
 
append using file2 
 
In we use the example of price data for crops, we would expect the variables to have the 
same name in the two files (i.e. crop, price). If the variables have different names, the 
resulting file will stack the variables with the same name and will create a new variable 
for the variable with a different name. For instance, suppose that in file 1 the variables 
were called crop and price and in file 2 crop and prices. The resulting dataset would 
contain three variables: crop, price, and prices. The user must therefore ensure that the 
resulting dataset is correct.  
 
append also has a number of other options which the user can learn about by typing help 
append. 

Chapter 2: String variables 
Getting to know the strings in your dataset 
 
This chapter will talk about the type of variable value we call “string,” but before we go 
on to that we should briefly review why this is important. The very first thing you want to 
do before starting to clean your dataset is getting to know your dataset in terms of the 
types of strings you have, and, more importantly, identify which variables you want as 
strings and which ones you do not. 
 
Use the describe command to see the types of variables that you have and what you need 
to work with. Two key types of variables you will use are numbers and strings.  
 
Numbers can take various formats (e.g. float, double, byte), but for now we will simply 
call them numbers. The main thing you need to watch for in terms of the number format 
is that when you convert a number into a string, it will keep the format of the number. So 
make sure you look at the number in the browse window and format it accordingly before 
turning it into a string.  
 
Strings are combinations of characters that are treated as "words," regardless of whether 
they are numbers, missing values, actual words or just random combinations of letters 
and numbers.  
 
Another important concept in Stata is labels. We often use numbers to describe categories 
in our data, and then assign labels to each number to more clearly identify (or, label) the 
values. When we assign labels to our numerical categories, however, if we browse our 
data (by using the command browse) Stata will show us our labels rather than our 
numbers. So, for example, if a variable gender actually takes on values 1 and 2, but is 
labelled "1 Female" and "2 Male", we will only see the Female and Male labels when we 
browse. Note these will generally be in a different colour than the strings or numbers 
(blue, red and black respectively). 

A quick aside on labels 
 
Since labels are a pretty important concept to grasp if you are going to turn numbers into 
strings and vice-versa, let’s do a quick recap of some of the main commands for labels.  
 
If you want to see what labels already exist in your dataset, use:  
 
label list  
 
This will list all the names of the value labels your dataset currently stores and what each 
number means.  
 
If you want to see the labels of a particular set of labels, then use  
 
label list [labelname], where [labelname] is the name of the label.  
 
You can also list all the label names by using label dir, though this will only show you 
the label names and not the label contents (as label list will). 
 

Creating the variables we need 
 
Creating field surveys is difficult, and there is usually a very limited amount of time the 
enumerator can spend with the person they are interviewing. There is also a limit on page 
space and data entry capacity. Thus, we often design surveys to be as easy as possible to 
conduct but also later allow us to create the variables we really want.  
 
For example, the enumerator may ask the respondent a question that has a scale of 1 to 5, 
where 1 is least happy and 5 is most happy. To make the enumerator’s life easier (s)he 
enters only the value 1, 2, 3, 4 or 5. After all, (s)he does not have the time to write down 
“least happy” or “most happy” 1000 times. In the raw dataset, then, we will have a 
variable with values of 1 through 5 and we know what these values mean (ie. we know 
their labels) from the questionnaire.  
 
A simpler and ubiquitous example is a variable that describes gender. It is normally 
coded as either a 1/0 or 1/2 variable, and the numbers mean male/female. As we’ve just 
discussed, this is the most efficient way of collecting this data. Since we have the 
questionnaire, we will know how to assign labels and clean this data.   
 
So let’s get cleaning. The first thing we want to do is describe our dataset, so we can see 
whether we have all the labels already assigned (if not, we’ll have to look to the 
questionnaire and assign them first), and then create new variables (or replace old ones) 
from numbers to strings. We will cover labelling in more detail in the chapter on 
numbers, but we will give a quick intro here for the purposes of the utility of labels in the 
context of strings. 

During cleaning, we generally create new variables rather than replace old ones, as it is 
very simple to drop variables that are not needed at a later stage. We might need a 
variable like gender to be described both in terms of numbers (ie. a 1-0 dummy variable) 
or a string (ie. male/female). We might want both of these to exist because during the 
analysis phase we need the number variable to be used in a regression, but it is easier to 
use the string variable to break up the dataset or use it as an option in graphs. 
 
With this in mind, one thing we may want to do is extract labels from our variables and 
make them into string variables themselves. This is done rather easily using the command 
decode. 
 
Continuing with our gender example, let’s say we have a variable called male that takes 
on the value of 0 if the individual is female, and 1 if the individual is male. We know 
what these 1s and 0s mean, so we need to label them.  
 
First define the label: label define male_label 0 “male” 1 “female” 
 
Then assign the label to the variable: label values male male_label 
 
Now we have a variable that is numeric, but has value labels. We now want to convert 
this to a string variable that will take on the labels as its string values. 
  
For example, if we want to create a variable that isn’t a binary male variable (ie. is or 
isn’t male), but one that fully describes the gender of the individual (in this case, male or 
female). 
 
We can do this with one simple command: decode. 
 
decode male, gen(gender2) 
 
What this is doing is taking the binary variable male and decoding it based on the value 
labels. We now have a variable, gender2, that takes on the value of “female” when male 
is 1 and “male “ when male is 0. 
 
Now let’s say we have the opposite. We have the variable gender that is a string but we 
want to have a numeric variable to describe the gender. What we want to do is encode 
our variable, and the code is: 
 
encode gender, gen(gender3) 
 
If you click on each of the labels (or use the codebook command to see it on the results 
screen) you would see that the encode command produces categorical variables starting at 
1. In this case, we have the label FEMALE on a value of 1, and MALE on a value of 2 
(the encode command sorts alphabetically first, so if you want to force a particular order 
it is useful to label values 1 value1 2 value2 etc). 

A quick aside on the encode command 
 
Note that the encode command will encode any number of values you have in a variable. 
It also encodes the variable based on alphabetical order, regardless of what order your 
data is actually in. That is, if you have a set of countries such as Brazil, Argentina, 
Uruguay and Bolivia (in this order), Stata will give values to them as such: 1-Argentina, 
2-Bolivia, 3-Brazil, 4-Uruguay. An example is shown in the pictures below. 
 

    
 
In the picture on the left, you see a variable, country, that has a random ordering of the 
four countries in our example. We then create a new variable, cty, using the encode 
command.  
 
encode country, gen(cty) 
 
As expected, the encode command creates a new numeric variable that assigned an 
ordinal number in alphabetical order to the different countries in our variable country. 
Remember that Stata shows numeric variables with value labels in blue, while strings are 
shown as red and numbers without labels are shown as black. For clarity of exposition, 
we have also added a variable cty2 that has the same values as cty but without the labels 
(here you can see how the alphabetical ordering works). 
 
Note how using encode to code dummy variables (which we will cover a bit later) is not 
the optimal solution, as you end up with 1 and 2 values instead of the 0 and 1 you would 
rather have. The encode command is useful when creating non-ordinal categorical 
variables that can have multiple values, such as a list of districts, countries, types of food 
etc. We will cover how to best create dummy variables later in this chapter. 

What are string variables? 
 
The word string in most programming languages generally refers to a string of characters. 
This could, of course, be an actual word, but it often is also very useful with numbers or 
alpha-numeric codes.  
 
String is a variable type, and with it comes a specific set of commands you can use as 
well as cannot use. For example: with string variables, you can trim, replace and insert 
characters in the string value of your variable. However, you cannot perform 
mathematical operations with a string variable even if it is a string of numbers (we will 
cover how to convert strings into number types later).  
 

Why are string variables useful? 
 
String type variables are useful for several reasons. First, it allows Stata to understand 
words and the user to program sets of commands given sets of words. Second, the string 
commands also allow users to quickly clean data entry errors, which abound in survey 
data. Finally, it is often very useful to transform numerical variables into strings in order 
to manipulate them before transforming them back into numerical variables. We will 
cover each of these in turn. 
 

A typical string variable 
 
Although it would be wonderful if all strings were inputted perfectly every time, this is 
seldom the case and so we must clean the string variables before we start working with 
them. Let’s start with a typical variable that is a string, such as a gender variable.  
 
individual_id and gender are the variable names, and the rows underneath are the values. 
 

 
 
If you type tab gender, you will get 

As you see, we have a serious issue with the classification of the variables. The only two 
categories we should see here are male and female, but we have several spellings of each. 
Thankfully, this is easily fixable with string commands.   
  

Cleaning string variables 
 
To clean string variables, the first thing we want to do is either put all values in upper or 
lower case. The commands are: 
 
replace gender=upper(gender) 
OR 
replace gender=lower(gender) 
 
Note that now we have a different set of variables to fix, since we have already taken care 
of different cases for the first letter of the word (such as the values “Male” and “male”). 
 

 
 
The next step is replacing misspelled words, which we have covered before.  
 
replace gender=”MALE” if gender==”MAEL” | gender==”M” 
replace gender=”FEMALE” if gender==”F” 

Now we only have two choices for our gender variable, as needed.  

Cleaning IDs 
 
Now we turn to cleaning ID variables. In our example, the variable name is individual_id. 
This ID has two letters in the front and three numbers, and it is divided by a space. 
 
Let’s say we don’t want a space in our ID. We can easily take it away using the command  
 
subinstr( )  
 
Type: 
 
replace individual_id=subinstr(individual_id,” “,””,.)  
 
The syntax of the command is: 
 
subinstr(s1,s2,s3,n)  
 
where: 

• s1: name of variable you want to change (alternatively, simply write your own 
string in this space) 

• s2: what is to be replaced 
• s3: what s2 will be replaced by 
• n: number of instances you would like to replace this (ie. If you only want the first 

space to be deleted, type 1, if you want the first two spaces, 2, etc. If you want all 
spaces to be deleted, then type a period: . ) 

 
In our case, we used the variable name in s1: individual_id, a blank space between 
quotes, “ “, for s2, and two quotes without a space between for s3, “”, and ended with a 
period, .   That is, we told Stata we want to replace all spaces with no space in the 
variable individual_id.  
 
Now we have: 

Now suppose we don’t need the identifiers “AC” in front of the numbers anymore. In that 
case, we can create a new variable, id2, that only uses the last three characters of 
individual_id. The command we use here is substr( ). 
 
generate id2=substr(individual_id,3,3)  
 
Here the syntax is: 
 
generate varname=substr(s1,n1,n2) 
s1 is the string we want to manipulate, in this case, individual_id 
n1 is the character position of the beginning of the new string to be built 
n2 is the length of the new string.  
 
In our example, we want our new variable to start in position 3 (where we know the first 
number is), and to be of length 3 (because we know all numbers are three digits). 
 

 
 
If we had IDs with different number lengths (something like AC1161), then we would 
instead want to use  
 
generate id2=substr(individual_id,3,.)  
 
In this case, we use a period as n2, which tells Stata to just go all the way to the end of 
the string. 
 
Another example would be if we wanted to create a new variable for gender that was only 
“F” or “M”, we could achieve this by either: 
 
generate gender2=”M” if gender==”M” 

generate gender2=”F” if gender==”F”  
 
OR, we could achieve this with one command: 
 
generate gender2=substr(gender,1,1) 
 
Because Stata do files can get quite large, it is always best to achieve your goals with as 
few lines of code as possible.  
 

Turning strings into number and vice-versa 
 
Consider the S4P2.dta file. We have an ID that is entirely made of numbers, and it is easy 
to turn it into string or back into numbers.  
 
To turn the id into string, type: 
 
tostring PRHS_2_ID, replace 
 
This will simply turn the PRHS_2_ID values into string values rather than number values.  
 
Alternatively, we could have created a new variable with the ID: 
 
tostring PRHS_2_ID, generate(PRHS_2_ID_string) 
 
To turn it back into a number, we use the destring command: 
 
destring PRHS_2_ID, replace 
 
OR destring PRHS_2_ID, generate(PRHS_2_ID_num) 
 

Creating dummy variables 
 
Now that we know what string variables are and how they can be used and manipulated, 
let’s consider how we can use these to create dummy variables. 
 
Dummy variables are useful when cleaning data, because we can check whether answers 
make sense.  
 
There are a few commands you can use to create dummy variables. Here are a few 
examples: 
 
tab variable, gen(dummyname) 
 
In our gender case we discussed earlier in the chapter, we would type: tab gender, 
gen(male) 

The result will be two new variables, male1 and male2. In the label, Stata automatically 
labels which variable is which. In this case, male1 is equal to 1 when gender is female, 
and male2 is equal to 1 when gender is male.  
 
Recode the variable as you want it: 
 
drop male1 
rename male2 male 
label variable male “=1 if gender=male” 
 
After creating the variable, make sure you tabulate the variable and check if it looks like 
what you wanted it to with tab male. 
 
Another way to create a dummy variable is by simply writing an “if” statement: 
 
generate male=1 if gender==”MALE”  
replace male=0 if male!=1 
 
This way we create a new variable and assign a value of 1 if the descriptive variable, 
gender, has value “MALE” and then replace any missing values with 0. 
 

Consistency Checks 
 
Consistency checks are exactly what the name implies: they are checks we do to ensure 
that the data is consistent with what we would expect to find across variables in the 
dataset.  
 
Let's consider the process of data collection. First, someone creates a field work form, 
then enumerators go to villages and cities and ask the questions necessary. They write all 
this down in these forms (or input them into electronic collection machines). Once this 
happens, it goes to a dataset that then needs to be cleaned.  
 
In this pathway, there is a lot of room for human error, and we need to ensure this is 
cleaned out of the data with the information that we have. We often use variables 
separately from others, so before we even start using the data we need to make sure they 
all make sense. For example, we will only do a tabulation of gender over households to 
see what the average ratio of male-female within households is, and we would only use 
the variable "gender" to do this, without including others such as "relationship to 
household head" or "age" in this particular tabulation. Thus, we need to first do 
consistency checks. 
 
Let's start with an example. We may have the household head being a man, and then 
three other members of the household: the wife, the son and the grandfather. We have 
data on both this relationship, and also the gender and age of each of these household 
members. Using our prior understanding of the subject country or region, we could 

expect the gender of the wife to be female, the son to be male, and the grandfather to be 
male. We could also expect the grandfather's age to be older than that of the household 
head, and the son to be younger than all of the other members. 
 
Which consistency checks we perform is variable, but there are usually quite a few. The 
very first one we should run is going through the questionnaire and seeing which parts of 
the questionnaire should only have been answered by a particular set of the respondents. 
For example, if there is an “eligibility question” at the beginning of a form, such as “Do 
you have any children under 2 years of age?” then if the answer is “yes” a further form is 
filled out, and if it is “no” then the enumerator is supposed to skip to the next section. 
These are quite common, and we can rather easily (though labouriously) go through all 
questions to ensure that a particular part of the dataset does not have respondents that 
should not be there.  
 
Continuing with this example, we would expect that the dataset related to the form about 
children under 2 years of age would only include children under 2 years of age. We can 
expect there to be all missing values for any individual who is over two years of age. We 
can then create a consistency check that ensures the variables in question only have 
answers for those individuals who are 2 years of age or under, and that we do not find 
data in these variables for individuals who are above two years of age.  
 
The second set of consistency checks are the ones we described above, such as 
“grandfather” being a male family member over a certain age. These should be agreed 
upon by the research team after careful consideration of the data at hand, the 
questionnaires, and the context of the country being studied. 
 
As with our prior example, with our dataset and prior knowledge, it is rather 
straightforward to ensure we have proper data if we systematically compare the gender 
variable with the household relationship variable, and the age variable with the same.  
 
This is the example we will pick up on in a second. Before we do that, however, let’s turn 
to what string variables are and how they can be useful. 

Checking data consistency 
 
Now let’s see which commands are useful to do consistency checks. After creating some 
extra variables (such as dummy variables), we can now use our variables to check the 
consistency of other variables. For example, using our prior knowledge we expect sisters 
to be female and brothers to be male. So how do we check to see if they’re consistent? 
 
In our example data, we have a new variable now: hh_rel, which we will call the 
relationship of the individual to the head of household. It can take on a series of values, 
including “Aunt”, “Brother”, “Grandfather”, “Sister”, “Wife.”  
 
Again using our prior knowledge, we would expect that brothers and grandfathers are 
male, and sisters, wives and aunts are female.  

To check this, we can use an “if” statement and our string variables. Clean the hh_rel 
variable in the same way we cleaned the gender variable. Then tabulate the hh_rel 
variable to know what values you need to check against. 
 
tab hh_rel 
 
Now we know what we have to classify. Create a new variable that will store the success 
or failure of the variable’s consistency, and write an if-statement to check this. 
 
gen check=. 
replace check=1 if male==1 & (hh_rel==”BROTHER” | hh_rel==”GRANDFATHER”) 
replace check=1 if male==0 & (hh_rel==”AUNT” | hh_rel==”SISTER” | 
hh_rel==”WIFE”) 
replace check=0 if check!=1 
 
tab check 
 
Then if there are some which are not consistent, check what the issues are: 
 
list gender hh_rel if check==0 
 
Then we know where the issues are we can go back to the rest of the data and fix it with 
the replace command. 
 

Manipulating string variables 
 
There are a set of commands that are very useful when manipulating string variables. We 
review the most useful here in a question and answer format.  
 
What do I do when I want to… 
 
…add characters to a string?  
 
Consider the following example for substituting all dots for commas in a numeric 
variable.  

replace variablename=subinstr(variablename, ”.” , ”,” ,.) 
 

 
 
If we only want to replace the first instance, then instead of a dot at the end use 1. For the 
first two instances, use 2 etc. 
 
replace variablename=subinstr(variablename, ”.” , ”,” ,1) 
 

 
 
replace variablename=subinstr(variablename, ”.” , ”,” ,2) 
 
…take away spaces? 
 
Sometimes we want to take away spaces from before or after a string value. There is a 
very easy way to do this in Stata: 
 
replace variablename=trim(variablename) 
 
…make all variables upper or lower case? 
 
As we mentioned in the case of cleaning variable values, it is always easiest to change 
values from its current case to either all upper or all lower case. That way, small 
variances such as “Brother” and “brother”, which would normally count as separate 
entries, will count as the same category (either “brother” or “BROTHER”). 
 
replace variablename=upper(variablename) 
OR 

replace variablename=lower(variablename) 
 
Alternatively, if you would like every letter to have a capital letter (for example, if you 
want the names of cities or people to look “proper”), Stata has a command for this: 
 
replace name=proper(name) 
 
…find a position of a letter within a string?  
 
Sometimes we want to find at what position within a string a particular part is. To find 
out the position and use the other commands to manipulate it, use the strpos command: 
 
generate position=strpos(variable,”string”) 
 
Where “string” is what you are looking for. Perhaps this is a period, or a particular letter, 
or even a word (such as “AND”). 

Chapter 3 – Numeric variables 
 
The previous chapter discussed the cleaning and the processing of string variables. This 
chapter deals with the core of Stata’s capabilities: numeric variables. Numeric variables 
are variables that express certain realities using numbers instead of text. To a large extent, 
household data contain numeric variables to express anything from age to income to 
gender to yield to levels of happiness. In this chapter, we explore some of the various 
dimensions involved in processing and cleaning household data using Stata. 
 
Researchers conducting household surveys often have to process the raw data received 
from the field.  The table below shows an excerpt of an actual unprocessed household 
dataset.  
 
Example of unprocessed household data 
 

Table 2 – Raw data 
 

 
 
Note: Names, gender and other information were changed to preserve privacy. 
 

Labeling numbers 
 
The first aspect of table 1 is that it is not immediately obvious what it contains. To 
understand it, it would be necessary to look at additional documents or the documentation 
of the data. Except for the household id and the names, it is not clear what the other 
variables are. For instance, RQ2 can be gender but it can also be whether the individual 
lives in a rural or urban area. Similarly, RQ4 can be the relationship to the head of the 
household but can also be whether the individual is employed, unemployed, looking for 
work or retired. 
 
Without further information, table 1 has very limited utility.  Two particular types of 
information would help us understand the table above: First, the definition of the 
variables and second, the meaning of the numbers. For instance, if RQ2 is gender, which 

PRHS_2_ID R_PID RQ1 RQ2 RQ3_YEARS RQ4
10101010101 1 JOHN 1 41 1
10101010101 2 MARY 2 41 2
10101010101 3 KACEY 2 21 3
10101010101 4 SUSAN 2 13 3
10101010101 5 RICHARD 1 12 4
10101010102 1 KELLY 2 41 1
10101010102 2 ANTHONY 1 37 2
10101010102 3 RUDOLPH 1 15 3
10101010102 4 PAUL 1 12 3

number is male and which is female? The label function in Stata, as we discussed in 
chapter 1, can help us address these two issues.  The syntax of the label command was 
shown in the first chapter. To identify what the variable contains, we can type: 
 
label var PRHS_2_ID “Household identifier” 
label var R_PID “Individual identifier” 
 
Variables can and should be labeled (as shown above) regardless of their type (string or 
numeric). However, only numeric values can be value-labeled.  When labeling a variable, 
ensure that the label used adequately describes the variable in a succinct but clear way. In 
the case of household data, labeling variables in a way that closely matches the questions 
in the questionnaire is often a good idea. Stata has a limit of 80 characters for labels and 
therefore conciseness is important when attributing a label. Should a label exceed 80 
characters, Stata will truncate it and, as a result, the label may become unclear or even 
not useful.  
 
Doing so would permit us to know what the variables contain by looking in the variable 
manager or by typing describe, which would return the follow result: 
 

 
 
 
The next step is to attribute labels to the categorical variables such as RQ2. RQ2 is a 
gender variable, so two value labels are appropriate: Male and Female.  Note, however, 
the two sexes are represented by 1 and 2. In this dataset, 1 represents Male and 2 
represents Female.  This classification is fine but may be confusing for the analyst, 
especially if he/she wants to run a regression using the gender variable.  Typically, this 
would be done using a dummy variable. While Stata can create dummy variables on the 
fly regardless of the classification, it may be more practical to define the gender variable 
as a dummy. 
 
This can be achieved in several ways, two of which are shown below. For simplicity, we 
assume that we want to measure the effect of having a male head of household. Thus, the 
dummy will defined as 1 for Males and zero for Females. 
 
replace RQ2=0 if RQ2==2 
 
Above, we replace the value 2 with zeros, thus creating a dummy. 
 
We could also have used the command recode, as in below: 
 
recode RQ2(2=0) 
 

R_PID           byte   %8.0g                  Individual identifier
PRHS_2_ID       double %11.0f                 Household identifier
                                                                                                                                            
variable name   type   format      label      variable label
              storage  display     value

Now that our gender variable is defined more accurately, value labels can be added. Note 
that to attribute value labels two commands are needed. The first one defines the label, 
the second one applies it to the relevant variables. The name of the variable defined in the 
first command can be anything, but often researchers use the name of the variable so that 
it is easy to remember. 
 
label define gender  1“Male” 0”Female” 
label values RQ2 gender 
 
If we tabulate our gender variable, we can see that the value labels have indeed been 
applied. 
 
tab RQ2 

 
 

Value-labeling numeric missing values 
 
Often, household data contain specific codes to indicate missing values or the reason for 
non-response. These codes are often specified as large negative values and, as a result, 
have to be handled with care in order not to affect the estimates of any subsequent 
estimations or analysis. It is easy to see how a code like -99 would, for instance affect the 
mean of a given variable. Suppose we have a dataset with 6 observations with the first 
one being a missing: .,4,5,2,1,6. The mean of all observations is 3.6. Now, consider the 
possibility that the first observation had been coded as -99. If left uncorrected, the mean 
would be -13.5 even though in reality there are no negative values in this set of data.  In 
addition to the mean, the standard deviation of the variable would also increase 
significantly from 2.07 to 41.  Such dramatic changes in the mean and in the standard 
deviation would render any analysis plain wrong. Even if regressions are not performed, 
keeping the -99 would provide completely wrong descriptives. 
 
However, coding variables with -99 does have the purpose to identify why the 
observation is missing.  In recognition of the many reasons for the existence of missing 
values, Stata has embedded in it an approach to deal with them in a way that identifies 
different categories of missing. 
 
Stata allows for 26 different categories of missing values which can be expressed using 
the letters of the alphabet followed by a dot (e.g. .a  .b  .c  etc).  Stata reads these codes as 
missing numeric values. This allows the researcher to keep track of the missing values of 
interest and at the same not risk contaminating the analysis. Labeling variables using 
letters of the alphabet is, as indicated, particularly useful for when codes (such as -99, -88 

      Total       13,377      100.00
                                                
       Male        6,774       50.64      100.00
     Female        6,603       49.36       49.36
                                                
     Gender        Freq.     Percent        Cum.

etc) were used to indicate the source of missing values. For instance, -99 may stand for 
not available and -88 may indicate that the respondent has passed away. 
 
The first step is to recode the variables that contain the negative values. 
 
recode x1 (-88=.a) (-99=.b) 
 
The second step is to apply a value label as was done previously. Suppose we were 
dealing with a gender variable. 
 
label define  gender 1”Male” 0”Female” .a ”Respondent died” .b ”Not available” 
 
Stata provides no specific guidelines on value-labeling missing values other than to 
indicate that the “size” of the missing value increases with the progression of letters, i.e.  
.a<.b<.c…<.z. However, it is often recommended that a systematic coding of missing 
values be applied throughout the various modules.  For instance, all observations that are 
missing due to lack of availability could be coded as .a. Similarly, observations for which 
no response was provided could be coded as .b.  Other codes can be established for  “not 
applicable” or “person has died” or any other possibility. An alternative possibility is to 
value-label missings according to the order of their magnitude, i.e. (-22=.a -44=.b etc). 
This has the advantage of making the general coding to clean missing values more 
consistent and easier to apply, but has the disadvantage of requiring negative numeric 
codes to be the same across modules. Unfortunately, numeric codes are rarely consistent 
across modules. In light of this reality, care must be taken when recoding and value-
labeling missing values across modules so as to avoid mis-labeling.  

Unique identifiers 
 
Household data often contain multiple modules covering a variety of different topics.  
Regardless of the topic, however, modules also have a specific unit of analysis (e.g. 
household, individual, crop, assets) based on level at which the analyst will conduct 
his/her analysis.  Once defined, the unit of analysis must be unique in order that 
observations belonging to a particular unit of analysis are represented only once. In table 
1, we notice that there are two identifiers, one for the household and one for the 
individuals of the household.  The unit of analysis is therefore the individual. Thus, the 
combination of household identifiers and individual identifiers must be unique, meaning 
there can be no repeats. 
 
To ensure units of analysis are unique, the command isid can be used. 
 
isid  PHRS_2_ID R_PID 
 
Lack of uniqueness can be a sign of data entry error or that the data are not in the 
appropriate format and thus need to be reshaped. The table below shows an example of 
the latter. 

Table 3 – Reshape 1 

 
 
The table shows data for households that were victims of shocks in their lives. Some 
households suffered more shocks than others and, as a result, household identifiers are 
not unique. Variable S9P2_Q9 reports the average financial loss caused by the shocks. 
Suppose, we would like analyze the magnitude of losses in conjunction with other 
household level variables. To combine these data with other data, we need to first ensure 
that the household identifier is unique. If the other dataset of interest also contained 
information on shocks, we could merge the two datasets using the household id and 
shock code.  In many cases, our interest would be to analyze it with other variables, such 
as income or education. Because these other datasets do not have a shock variable, we 
cannot merge them without first making the household id unique. 
 
To do so, we need to reshape wide the data in the table. We need to generate one 
variable for each type of shock and then stack the observations of financial losses under 
each of the relevant shocks. For simplicity, we will create one variable y_shock_i, where i 
corresponds to the shock’s id. 
 
First, we generate a string variable out of the shock id. 
 
tostring S9P2_CODE, gen(S9P2_CODESTR) 
gen  S9P2_CODEST=”y”+ S9P2_CODESTR 
reshape wide S9P2_Q1 S9P2_Q9, i( PRHS_2_ID) j( S9P2_CODEST) string 
   

PRHS_2_ID S9P2_CODE S9P2_SHOCKS S9P2_Q1 S9P2_Q9
10101020204 38 Public Fights between individuals Yes 10000
10101020204 39 Clashes/ fights between families Yes 0
10101020204 51 Firing to intimidate Yes 0
11127414214 39 Clashes/ fights between families Yes 200000
11127414214 40 Clashes/ fights between baradaris Yes 20000
11127414214 41 Clashes/ fights between caste groups Yes 30000
11229444503 39 Clashes/ fights between families Yes 15000
11229444503 40 Clashes/ fights between baradaris Yes 6000
11229444503 41 Clashes/ fights between caste groups Yes 1700
11231474821 39 Clashes/ fights between families Yes 0
11231474821 40 Clashes/ fights between baradaris Yes 0
11231474824 38 Public Fights between individuals Yes 1500
11231474824 39 Clashes/ fights between families Yes 0
21434525325 39 Clashes/ fights between families Yes 0
21434525325 41 Clashes/ fights between caste groups Yes 0
21744646525 40 Clashes/ fights between baradaris Yes 0
21744646525 53 Violent public protests/ riots Yes 0

Table 4 - Reshape 2

 
 
Due to space reasons, we have only presented the reshaped table for two shocks. Shocks 
38 and 39 correspond to public fights between individuals and clashes/fights between 
families. Key in the table above is the fact that the PRHS_2_ID is now unique and can be 
merged.  The reader can verify that the variables were created properly.  
 
Note that there is a user-written command that allows labels to be retained called 
reshape8. It can be installed via Stata’s user interface and used in the same way as 
reshape. More information is in the documentation of the program through Stata’s help 
function. 
 

Consistency checks 
 
Once appropriate variables are labeled, we can proceed to perform consistency checks on 
variables of interest. Household data combine a series of variables that have logical 
relationships with one another. As a result, it is recommended that these relationships are 
assessed to ensure correctness. For instance, in the table below we show such a 
relationship. 
 

Table 5 – Consistency checks 

 
 
The relationship in question is between RQ2 (gender) and RQ4 (relationship with the 
head of the household). The variable RQ4 specified the categories of the relationship but 
also provide additional indication of the gender. Thus, one important consistency check is 

PRHS_2_ID S9P2_Q1y38 S9P2_Q9y38 S9P2_Q1y39 S9P2_Q9y39
10101020204 Yes 10000 Yes 0
11127414214 Yes 200000
11229444503 Yes 15000
11231474821 Yes 0
11231474824 Yes 1500 Yes 0
21434525325 Yes 0
21744646525
31947727314

R_PID RQ1 RQ2 RQ3_YEARS RQ4 Code RQ4
1 JOHN male 41 head 1
2 MARY female 41 wife 2
3 KACEY female 21 daughter 3
4 SUSAN female 13 daughter 3
5 RICHARD male 12 son 4
1 KELLY female 41 niece 5
2 ANTHONY male 37 uncle 6
3 RUDOLPH male 15 son 4
4 PAUL male 12 son 4

to ensure that all the female categories (daughter, wife and niece) are indeed female. 
Conversely, all male categories are male.  Consistency checks can be performed in a 
number of ways. Below we list a few. Using the assert command: 
 
assert RQ2==1 if RQ4==1 
 
We can also perform the test for multiple entries using the command inlist:  
 
assert RQ2==1 if inlist(RQ4,1,4,6) 
 
Note that in the example above the assert command is composed of two logical 
statements as is indicated by the presence of the two equal signs both before and after the 
“if” condition. Using the count command: 
 
count if RQ2==1 & inlist(RQ4,2,3,5) 
 
Above Stata would count the occurrences of males for the categories of wife, daughter 
and niece. The analyst would therefore expect to see a zero as a count. The reverse can be 
done for the females. Using the tabulate command: 
 
tab RQ4 if RQ2==”Male” :gender 
 
The tabulate command will present a tabulation of all categories of RQ4 for individuals 
classified as Male. The analyst would therefore confirm that no females are present. 
A small Stata trick was employed in the tabulate command line. If variables have many 
categories, it may be hard to remember all the numbers for all the categories. Using the 
trick above, we can just enter the label value as long as we specify the label name after a 
colon.   

Dealing with missing variables 
 
Missing observations are extremely frequent occurrences in household data.  Handling 
missing values is a necessary and important step in the data cleaning process. The starting 
point is to understand how Stata “sees” missing values. 
 
As discussed in chapter 1, missing values are treated by Stata as very large numbers. 
Stata also ignores rows of data that contain missing observations. In a regression setting, 
therefore, the number of observations actually present in the regression may be a lot 
lower than univariate descriptive statistics may suggest. 
 
In the process of generating variables, the analyst needs to ensure additional missing 
values are not created via transformations in the data. A transformation that immediately 
comes to mind is the natural log in the presence of zeros.  Because the log of zero is not 
defined, Stata will replace the ln(0) with a missing value. Doing so, however, will 
introduce a bias in the data because the pattern of missing observations will become 
systematic. A simple trick to avoid this is displayed below.  

generate lyield= ln(1+yield) 
 
This is merely a computational exercise, however. The issue of how to deal with missing 
values in your analysis is much more complicated and requires thought and contextual 
analysis to decide the best approach. We merely point to how to compute a solution 
which has sometimes been used and do not mean to imply this is the best way to deal 
with zeroes in your data.  
 
Issues with variable creation involving missing observations are not limited to 
transformations. Performing simple operations such as an addition can be significantly 
impacted by the presence of missing. For instance, if we want to obtain the sum of two 
variables using generate but one of them has missing values, the results will be missing 
for the rows that contain missing values. To obtain a valid sum, the command egen needs 
to be used.  
 
egen x3=rowtotal(x1 x2) 
 
The function rowtotal treats missing values as zeros and in doing so is capable of 
performing the correct summation. 
 
egen has a number of other row functions that also treat missing as zeros. Some of them 
are:  
 
rowmean- calculates the mean 
rowmedian-calculates the median 
rowmax-returns the maximum (non-missing) observation 
rowmiss- counts the number of missing 
rownonmiss- counts the number of non-missing 
rowpctile- returns the desired percentile (specified in an option) 
rowsd- returns the standard deviation 
  

Descriptive Statistics as a tool to understanding your data 
 
Descriptive statistics can be a very important tool in the process of data cleaning. By 
tabulating data and obtaining relevant measures of dispersion, it is possible to 
immediately identify outliers and other potential problems. In the table below, we provide 
hypothetical yield data as an example. 

Table 6 – Descriptive stats 

 
 
 
In the table we have entered one measure of yield that is in a different unit just to 
demonstrate how descriptives can point us to problems. Suppose we were not aware of 
the unit problem and decided to calculate the means of yields for the three crops. 
 
tabstat yield, by(crop)  
 
 
 

 
 
The table above shows that wheat has a much greater yield than the other two crops. 
Given the discrepancy in the levels of magnitude, it is prudent to look into the wheat 
yield more carefully.   
 
summ yield if crop==”Wheat” 
 

 
Now we can be certain that indeed there is a problem in the yield variable. The standard 
deviation is very high compared to the mean and the maximum value is 800.  With this 
information in hand, it is possible to find the entry and check for units or data entry 
errors. 
 
While the example above is very simple it illustrates how basic descriptive statistics can 
be helpful in identifying outliers and data problems. 

crop yield unit plot
Wheat 6 tons per ha 1
Rice 6 tons per ha 1
Other 1 tons per ha 1
Wheat 1 tons per ha 2
Rice 2 tons per ha 2
Other 3 tons per ha 2
Wheat 4 tons per ha 3
Rice 4 tons per ha 3
Other 4 tons per ha 3
Wheat 800 kilos per ha 4
Rice 4 tons per ha 4
Other 12 tons per ha 4

                  
 Wheat      202.75
  Rice           4
 Other           5
                  
  crop        mean

       yield           4      202.75     398.172          1        800
                                                                      
    Variable         Obs        Mean    Std. Dev.       Min        Max

Outlier detection 
 
More general outlier detection procedures can be applied to detect outliers in continuous 
variables. Below we describe one of them.   
 
The range method can be best explained by examining a box plot. 
 

Figure 4 – Box plot 

 
For the purposes of the graph, a randomly uniform distributed variable was generated in 
Stata. The variable ranges from 0 to 1 but these magnitudes are not relevant to the 
strategy. What is relevant, however, is the position of the horizontal bars in the graph. 
The bottom and the top horizontal bars represent the lowest and highest values in the 
distribution. The box in the middle is framed by the first quartile (q1) at the bottom and 
the third quartile (Q3) at the top.  The line in the middle of the box is the median. The 
graph thus provides a general but accurate view of the distribution. Points outside the 
bottom and top line of the graph are outliers. In cases of very large outliers the box gets 
compressed in the bottom and dots appear on the top of the graph.  
 
The range method identifies outliers via the following formula: 
 
For low-value outliers: an observation j in variable x  is an outlier if  xj<Q1-1.5IQR, 
where IQR is the interquartile range, which is given by Q3-Q1. 
 
For high value outliers:  an observation j in variable   is an outlier if  xj>Q3+1.5IQR, 
where IQR is the interquartile range, which is given by Q3-Q1. 

The distance between Q1, Q3 and the bottom and top lines is given by 1.5*IQR.  Hence 
values that are smaller (for the bottom part) or greater (for the top part) than this distance 
are outliers. 
 
The method allows for the identification and their subsequent removal or substitution. 
The figures below illustrates the effectiveness of identifying and removing outliers using 
this method.  

Figure 6 – Box plot with and without outliers 

 
The range method requires the calculation of the first and third quartile and the inter-
quartile range. These calculations can be performed in Stata using the code below: 
 
foreach var of varlist x1 x2 x3{ 

generate out`var'=. 
summarize `var',detail 
replace out`var'=1 if `var'>r(p75)+(1.5*(r(p75)-r(p25))) & `var'!=. 
order out`var', after(`var') 
label var out`var' "Values marked 1 in variable `var' are outliers" 

} 
 
The code displayed above draws from the Stata’s ability to store results after estimation 
commands.  It also uses Stata’s looping command to apply the procedure to a number of 
variables. The first line of the code indicates to Stata which variables are to be examined 
in the loop. The second line generates new variables that will contain an indicator 
variable of whether the observation is an outlier. Note that the newly generated indicator 
for every variable is initiated as being missing. This is because it will be replaced in a 
subsequent step. The third line summarizes the data. The option detail tells Stata to 
provide a full summary of the data, including the percentiles.  The replace statement then 
applies the formula using the values stored after the summarize command. The two 
subsequent commands just order and label the variables respectively. 

Dealing with outliers 
 
Many researchers choose to replace the identified outliers by the mean of other (non-
outliers) observations. However, other methods of imputation are available, including: 1) 
replacing the missing values with the latest available observation (this is more useful 
when time-series data are available), and 2) using a multiple imputation strategy. Stata 
offers a complete suite on multiple imputation and the user is encouraged to refer to the 
manuals for a description of it. Because a number of assumptions have to be in place to 
apply multiple imputation, we do not present its syntax or commands here in order to 
avoid misleading information.   
 
One distinction among variables was made before applying the procedure. I identified 
variables that were related to specific crop categories and variables that were general to 
households. Crop variables like crop area and crop quantity were corrected within the 
crop category they belonged to. Other variables were corrected across all households, 
irrespective of the crop. 

Chapter 4: A strategy to clean data 
using Stata 
 
In the previous chapters we have covered a number of topics that are relevant to data 
cleaning. At the outset, a basic introduction to Stata was given to familiarize the user with 
Stata commands. Then, a guide to string and numeric variable was provided, which 
illustrated the methods and capabilities that can be employed in data cleaning by using 
Stata.  This chapter complements the previous chapters by providing an overview of a 
general strategy to conduct data cleaning. In particular, we outline the key steps and their 
sequence involved in cleaning household data. We also introduce the reader to Stata’s 13 
new project manager capabilities, which can be a very helpful tool in organizing the files 
and do-files. The project manager capability allows us to keep things neat and tidy. 
Lastly, we offer some advice on how to present the final product (a clean dataset) and 
what additional documents should also be provided. 
 
In the table below we provide a sequence of steps to be taken in order to obtain a clean 
dataset. However, prior to doing so, let us be reminded of an important step that has to be 
always present: hard copy checks.  Prior to conducting changes in Stata, any suspicious 
data point in the dataset should be checked against the hard copies of the surveys. 
Suspicious data include duplicates in identifiers, misspelled names, missing information 
that should not be missing (for instance, gender or issues that may arise from consistency 
checks) and finally outliers.  
 

Table 7 – Cleaning data strategy steps 
Order Step Relevant commands Reason 
1 Label/re-label 

variables and label 
their values 

label (variable and 
value labels).  
recode if needed to 
convert negative 
codes to letters. 

Identifies the 
variables, deals with 
missing information 
and strange codes. 

2 Check for unique 
identifiers 

isid, codebook, 
inspect, duplicates 

Essential to perform 
correct merges, 
identify individuals 
and reshape data 
correctly 

3 Consistency checks assert, tabulate, 
summarize, table, 
tabstat, count 

Ensures logical 
connections are 
present in the data, 
e.g. mothers are 
females. Also 
checks for 
completeness of the 
dataset e.g. to verify 
that all quantities 

have units.  
4 Correct string 

variables 
replace, substr, 
subinstr, index 

Correct spellings, 
make the data 
uniform. 

5 Ensure relevant 
households/members 
are present e.g. 
males 16 years old 
or older 

assert, count 
 

Complements 
consistency checks.  

6 Reshape variables to 
appropriate format 

reshape 
 

To be done on an 
as-needed basis if 
merges need to be 
performed.  

7 Merge merge 1:1 
merge m:1 
merge 1:m 

Combines datasets 

8 Document steps in 
the do-file (see 
discussion below) 

Use do-file editor Ensures steps can be 
replicated and serve 
as reminders to the 
researcher as to why 
such actions were 
taken.  

9 Prepare appropriate 
documentation (see 
discussion below) 

Use a word-
processor 

Identifies the 
datasets, issues and 
steps taken. 

10  Use the project 
manager tool (see 
discussion below) 

Do-file menu – on 
Stata 13 only 

Organizes files of 
the project, making 
it easy to find, make 
changes and 
document. 

 
Steps 1-7 were discussed in the previous sections.  The three remaining steps are 
discussed below 
 
Step 8 essentially consists of transforming your do-file in an essential component of the 
entire process, by making it a document to which you and people around you can 
immediately identify and replicate the steps and actions taken in the data cleaning.  Stata 
makes documenting a do-file an easy process by skipping any lines that are preceded by 
an  asterisk and treating these as comments.  An example of a documented do-file is 
shown below: 

************************************************************************ 
*Education Module – S1P1 
*Level of identification: Household and Individual 
*Outcome files: s1_hh s1_final 
************************************************************************ 
cd C:\Users\Eduardo\Dropbox\HH_PSSP\malestata 
 
use education, clear 
*ensuring identifier name is consistent with other files 
ren edu_pid r_pid 
isid prhs_2_id r_pid 
 
*Below we call the label files 
qui do label_fix_all 
qui do labels_s1p1 
 
* consistency checks 
 
* checking that individual that attended school have valid school information 
assert s1q7!=. if s1q5==1,rc0 
assert s1q8!=. if s1q5==1,rc0 
 
* checking that individuals that do not attend school don't have a school name 
assert s1q7==. if s1q6!=.,rc0 
assert s1q10!=. if s1q9==0,rc0 
assert s1q10==. if s1q9==1,rc0 
assert s1q11!=. if s1q9==0,rc0 
assert s1q12!=. if s1q9==1,rc0 
* when merged with Roster, need to check ages>19 
merge 1:1 prhs_2_id r_pid using 
C:\Users\Eduardo\Dropbox\HH_PSSP\malestata\core\roster_final, 
keepusing(rq2 rq3_years)  
keep if _merge==3 
summ rq3_years if s1q5==1 
* ages are ok 
assert rq2==1 if s1q5==1,rc0 
drop  _merge 
order rq2 rq3_years, after(r_pid) 
sort prhs_2_id r_pid 
saveold  s1_individual,replace 
 
As can be seen from the example, the do-file starts by providing the user with 
information about the file and the output files that the do-file will generate. Note also that 
all the checks that are made are preceded by a comment indicating what is being achieved 
with the checks. Finally, comments can also be placed to indicate that the data are okay 
(or have problems). 

Preparing the appropriate documentation is also an essential step in the data cleaning 
process. The documentation should contain a list of all the datasets, the procedures 
applied to each of them (e.g. labeling, reshaping, merging) and a complete list of do-files 
used in the project. While this may seem rather tedious, it really pays dividends in the 
mid to long run, especially if others are going to use the data you cleaned.  Box 1 
describes the key elements of a functional documentation process. 
  

Box 1: Documenting your data cleaning. 

An effective documentation should inform the reader of all the steps that were taken during 
the data cleaning in clear and concise way. Here, we provide some of the important 
elements that should be part of your documentation. 

1. A complete list of all the files for which cleaning was performed:  It is essential 
to inform the user of extent of the data and of the cleaning efforts. Household 
surveys contain many modules and thus providing a list by module is helpful and 
often helps future users to know understand the data better. 

2. Identify and list the do-file(s) used during the cleaning and the purpose they 
served. For example, are they used for labeling, merging, generating tables, 
consistency checks etc. 

3. Clearly identify the steps taken in the process of cleaning the data: The 
documentation along with your do-file should serve as a map that takes the user 
from the raw data to the finished product. 

4. Inform all changes that were conducted during the cleaning: The first set of 
important changes are those that originated from checks on the hard copies. These 
changes which can range from spelling corrections to correction in the identifiers to 
outlier checks and corrections should be clearly indicated in the documentation.    

5. Issues regarding to the merging of files should be highlighted and provide the 
necessary steps taken to perform the merger, particularly if errors occurred or 
observations were left without a match. 

6. List corrections that were performed based on the input of researchers or the 
data collectors: For instance, often enumerators will build a relationship with the 
communities they visit and will gain significant understanding of the region. In 
some cases, they may be able to provide information or corrections that were not 
actually part of the hard copy checks. Similarly, researchers may choose to 
eliminate outliers in particularly way. Whatever the source of change is it should be 
clearly indicated in the documentation. 

The last item mentioned in the table refers to the project manager tool, which became 
available with the release of Stata 13. A screenshot of the manager used to clean a dataset 
has been pasted to the next page. The project manager is accessed via the do file editor, 
where the user can start a new project or open and modify an old one. Once a new project 
is given a name, the user can easily add groups (folders) to the project.  Within folders, 
users can add a new file or add existing files to each group. In the next page, a folder for 
Agriculture was created. Within agriculture, we see that there two sub-folders: Labels and 
Raw data. Each of these folders contain the relevant do files or Stata files. The content of 
these files can be changed just by double-clicking on the item. Stata’s project manager 
allows for many do-files to be open at the same time, making cross-checking across do-
files easy. Users can also run the do-files open in the project manager. 
 
The project manager along with the documentation and organized do-files can greatly 
improve the presentation and organization of the output to be delivered in the data 
cleaning. Being specific about what each do-file does is also very helpful, so naming do-
files appropriately is essential. In the project manager screenshot, we notice that all label 
files are named labels-module.do, indicating that labels are for specific modules.  Stata’s 
ability to call other do-files within a do-file can also greatly improve the neatness and 
organization of the product.  By typing do  do_file_name , we can easily run parts of the 
cleaning (such as labeling)  that do not need to be in the same do-file. Thus, we can have 
one do-file that performs all the main actions in the cleaning for a given module and one 
that does just the labels. 
 
Cleaning data is long and often protracted process. Stata can significantly help in the 
process by providing tools that are easy to use, to document and to analyze data. As 
powerful as it may be, Stata does not replace the keen eye of the analyst that is 
performing the cleaning.  
 
Think of Stata as your very powerful hammer. You still need to painstakingly look 
through the documentation of the survey and ensure the data you are looking at matches 
the data that was supposed to be collected. That is, carefully perusing the questionnaires 
and checking that the data makes sense across rounds, keeping track of the number of 
observations you are expecting to have etc.  
 
Keep in mind once a dataset is proclaimed clean, many researchers will use that data to 
conduct important research and much of this is to inform public policy. It is a big 
responsibility to clean datasets, and we hope this manual can help you in doing this job 
with the care and concern it deserves. 

Figure 6 – Stata 13 Project Manager 

